// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (access/Ownable.sol)

pragma solidity ^0.8.20;

import {Context} from "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * The initial owner is set to the address provided by the deployer. This can
 * later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    /**
     * @dev The caller account is not authorized to perform an operation.
     */
    error OwnableUnauthorizedAccount(address account);

    /**
     * @dev The owner is not a valid owner account. (eg. `address(0)`)
     */
    error OwnableInvalidOwner(address owner);

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the address provided by the deployer as the initial owner.
     */
    constructor(address initialOwner) {
        if (initialOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(initialOwner);
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        if (owner() != _msgSender()) {
            revert OwnableUnauthorizedAccount(_msgSender());
        }
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        if (newOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/Context.sol)

pragma solidity ^0.8.20;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/SimplySendIt.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.20;

import {Ownable} from "@openzeppelin/contracts/access/Ownable.sol";

contract SimplySendIt is Ownable {
    // error codes
    error ZeroValue();
    error InsufficientBalance();
    error EmptyReceivers();
    error EmptyAmounts();
    error LengthsNotEqual();
    error FailedToSendEther();
    error MaxWalletsLengthExceeded();
    error InvalidFeePercentage();

    // events
    event Received(address indexed sender, uint256 amount);
    event WithdrawnBundle(address[] indexed receivers, uint256[] amounts);

    // state variables
    uint8 public projectFee;
    uint8 public distributorFee;
    address private projectAddress;
    address private adminAddress;

    modifier onlyAdmin() {
        require(adminAddress == _msgSender(), "Admin: caller is not the admin");
        _;
    }

    constructor(
        address initialOwner,
        address _projectAddress,
        address _adminAddress
    ) Ownable(initialOwner) {
        projectFee = 1; // 1%
        distributorFee = 1; // 1%
        projectAddress = _projectAddress;
        adminAddress = _adminAddress;
    }

    // receive eth function
    receive() external payable {
        if (msg.value == 0) revert ZeroValue();
        sendEth(adminAddress, (msg.value * distributorFee) / 100);
        sendEth(projectAddress, (msg.value * projectFee) / 100);
        emit Received(msg.sender, msg.value);
    }

    // set fee percentage, only owner can call this function
    function setProjectFee(uint8 _projectFee) external onlyOwner {
        if (_projectFee > 99) revert InvalidFeePercentage();
        projectFee = _projectFee;
    }

    function setDistributorFee(uint8 _distributorFee) external onlyOwner {
        if (_distributorFee > 99) revert InvalidFeePercentage();
        distributorFee = _distributorFee;
    }

    function setAdminAddress(address _adminAddress) external onlyOwner {
        adminAddress = _adminAddress;
    }

    function setProjectAddress(address _projectAddress) external onlyOwner {
        projectAddress = _projectAddress;
    }

    // withdraw eth to multiple, only owner can call this function
    function multiWithdraw(
        address[] memory receivers,
        uint256[] memory amounts
    ) external onlyAdmin {
        // check if receivers and amounts are not empty
        // receivers and amounts have same length
        // and receivers length is less than 200
        if (receivers.length == 0) revert EmptyReceivers();
        if (amounts.length == 0) revert EmptyAmounts();
        if (receivers.length != amounts.length) revert LengthsNotEqual();
        if (receivers.length > 200) revert MaxWalletsLengthExceeded();

        // calculate total amount and check if contract has enough balance
        uint256 totalAmount = 0;
        for (uint i = 0; i < amounts.length; i++) {
            totalAmount += amounts[i];
        }
        if (totalAmount > address(this).balance) revert InsufficientBalance();

        // send eth to receivers
        for (uint i = 0; i < receivers.length; i++) {
            sendEth(receivers[i], amounts[i]);
        }
        emit WithdrawnBundle(receivers, amounts);
    }

    function sendEth(address receiver, uint256 amount) internal {
        (bool sent, ) = payable(receiver).call{value: amount}("");
        if (!sent) revert FailedToSendEther();
    }
}